

STANOVENÍ ZBYTKOVÉHO VÝKONU JADERNÉHO PALIVA

DETERMINATION OF THE RESIDUAL POWER OF THE FISSIONABLE FUEL

Stanislav BARTOŇ, Pavol TANUŠKA

Autori: Doc. RNDr. Stanislav Bartoň, CSc., Doc. Ing. Pavol Tanuška, PhD.
Pracoviisko: Ústav aplikovanej informatiky, automatizácie a matematiky,
Materiálovotechnologická fakulta STU
Adresa: Paulínska 16, 917 24 Trnava
Email: barton@mendelu.cz, pavol.tanuska@stuba.sk

Abstract

V článku jsou rozvedeny matematické základy modelu sdílení tepla a ochlazování kontejneru s vyhořelým jaderným palivem. Z rozboru modelu je navržena metodika měření rozhodujících fyzikálních veličin a následně způsob výpočtu zbytkového výkonu.

Principles of the mathematical modelling of the heat exchange and cooling of the container with spent fission fuel are explained in the presented article. Methodics of the measurement of the main physical quantities and algorithm of the computation of the remaining power are based on this model.

Key words

Přestup tepla, tepelná kapacita, koeficient přestupu tepla, matematické modelování, nelineární metoda nejmenších čtverců, Maple

Heat exchange, heat capacity, heat transfer coefficient, mathematical modelling, Non-linear least squares method, Maple

Úvod

Palivo po vyjmutí z reaktoru není zcela beze zbytku vyhořelé, tzn. že v něm stále ještě probíhají jaderné reakce. Důsledkem těchto reakcí je mimo jiné i vznik tepelné energie, která ohřívá celý palivový článek. Pokud by se článek neochlazoval, zvyšovala by se jeho teplota a v případě, že by přesáhla konstrukční limity, došlo by k jeho destrukci. Znalost zbytkového výkonu je proto důležitá pro bez-

pečné skladování vyhořelého jaderného paliva.

Pomocí programu Maple se sestaví matematický model popisující časovou závislost teploty vnější stěny kontejneru na zbytkovém výkonu paliva, způsobu ochlazování a okolních podmínkách. Tento model bude možné použít i inverzně – ze známé teploty kontejneru, způsobu ochlazování a okolních podmínek lze určit zbytkový výkon paliva.

Vstupní příkazy a odpovědi programu Maple budou sázeny následujícím způsobem

```
> e1 := Int(x, x=a..b) = int(x, x=a..b);
```

$$e1 := \int_a^b x \, dx = b^2 - a^2$$

Matematický model

Pro kontejner s vyhořelým jaderným palivem je možné zapsat zákon zachování energie, viz např. [1], ve tvaru:

```
> restart;
> e1 := P*t = K*(T(t)-Ti) + Int(S*alpha*(T(t)-Te), t);
```

$$e1 := P t = K (T(t) - T_i) + \int_0^t S \alpha(T(t)) (T(t) - T_e) dt,$$

kde

P = zbytkový výkon	T = okamžitá teplota povrchu
t = čas	S = plocha povrchu kontejneru
K = tepelná kapacita kontejneru s palivem	α = koeficient přestupu tepla, obecně závisí na teplotě povrchu
T_i = počáteční teplota povrchu	T_e = teplota okolí

Integrál z rovnice e1 je možné odstranit derivací této rovnice podle času a z výsledku vyjádřit derivaci teploty podle času $dT(t)/dt$

```
> e2:=diff(e1,t);
```

$$e2 := P = K \frac{d}{dt} T(t) + S \alpha(T(t)) (T(t) - T_e)$$

```
> e3:=diff(T(t),t)=solve(e2,diff(T(t),t));
```

$$e3 := \frac{d}{dt} T(t) = -\frac{S \alpha(T(t)) (T(t) - T_e) - P}{K}.$$

Rovnice e3 je obyčejnou diferenciální rovnicí pro časovou závislost teploty, kterou ovšem výrazně komplikuje obecně neznámá závislost koeficientu přestupu tepla na teplotě $\alpha(T(t))$. Nicméně pomocí programu Maple je možné nalézt její řešení.

```
> dsol1:=dsolve(e3,T(t));
```

$$\text{dsol1} := t - \int^{T(t)} \frac{K}{-S\alpha(-a) - a + S\alpha(-a) T e + P} d_{-a} + _C1 = 0. \quad (1)$$

Maple našel řešení rovnice e3 ve tvaru inverzní funkce $t(T)$, navíc je nutné znát funkční závislost $\alpha(T(t))$, bez kterého nelze v analytickém řešení rovnice (1) dále pokračovat.

Vraťme se proto k rovnici e3. Vzhledem k tomu, že koeficient přestupu tepla $\alpha(T(t))$, je neznámou funkcí teploty, ale teplota samotná je funkcí času, její funkční závislost také samozřejmě neznáme, je možné použít substituce $\alpha(T(t)) = \beta(t)$, čili konstatovat, že koeficient přestupu tepla je neznámou funkcí času. Rovnice e3 tak přejde na tvar jednodušší diferenciální rovnice, pro kterou již lze nalézt analytické řešení.

```
> e4 := subs(alpha(T(t))=beta(t), e3);
```

$$e4 := \frac{d}{dt} T(t) = - \frac{S \beta(t) (T(t) - T_e) - P}{K}$$

```
> dsol2 := simplify(dsolve(e4,T(t)));
```

$$\text{dsol2} := T(t) = \frac{\int e^{\frac{S}{K}} \int \beta(t) dt (S \beta(t) T_e + P) dt + _C1 K}{K e^{\frac{S}{K}} \int \beta(t) dt}. \quad (2)$$

Stále zůstává problémem neznalost závislosti koeficientu přestupu tepla na čase, nicméně rovnice (2) již obsahuje přímý vztah pro $T(t)$. V dalším kroku se pokusíme určit integrační konstantu $_C1$.

Nejprve je nutné doplnit ve všech integrálech integrační meze. Dále je možné předpokládat, že v průběhu měření se teplota stěny kontejneru bude pohybovat pouze v úzkém rozmezí, díky tomu je možné nahradit neznámou funkci závislosti koeficientu přestupu tepla na čase, *nebo na teplotě*, střední hodnotou

$$a = \frac{\int_0^t \beta(t) dt}{t} = \frac{\int_{T_i}^{T_f} \alpha(T) dT}{T_f - T_i}.$$

```
> T1 := T(t) = (Int(exp(1/K*S*Int(beta(t),t=0..t)) *
> (S*beta(t)*Te+P),t=0..t)+_C1*K) *
> exp(-1/K*S*Int(beta(t),t=0..t))/K;
```

```
> T2 := subs(beta(t)=a, T1);
```

$$T2 := T(t) = \frac{\int_0^t e^{\frac{S}{K}} \int_0^t a dt (S a T_e + P) dt + _C1 K}{K e^{\frac{S}{K}} \int_0^t a dt}. \quad (3)$$

Rovnici (3) je možné vyjádřit v analytickém tvaru

```
T3:=collect(expand(value(T2)), exp);
```

$$T3 := T(t) = T_e + \frac{P}{aS} + \frac{-T_e - \frac{P}{aS} + _C1}{\frac{aS}{K}e^{\frac{t}{\tau}}} \quad (4)$$

Integrační konstantu $_C1$ určíme z podmínky, $\lim_{t \rightarrow 0} T(t) = T_i$, teplota stěny kontejneru v čase $t = 0$ je rovna počáteční teplotě T_i . Jednoduchý výpočet vede k výsledku $_C1 = T_i$.

Podobným výpočtem pro finalní ustálenou teplotu stěny kontejneru obdržíme rovnici pro konečnou teplotu stěny kontejneru

```
> Tf = expand(limit(rhs(T3), t=infinity))
> assuming S>0, K>0, a>0;
```

$$T_f = T_e + \frac{P}{Sa}.$$

Protože ustálenou teplotu stěny kontejneru T_i i teplotu okolí T_e lze přesně změřit, je možné při známém topném výkonu P , který je možné provést například elektrickým ohřevem, určit součin aS .

```
> map(u->(u-Te)*S*a/(Tf-Te), %);
```

$$aS = \frac{P}{T_f - T_e}.$$

Rovnici (4) je tak možné pomocí substituce

```
> su := \_C1=Ti, P=C1*S*a, K=tau*S*a;
```

$$su := _C1 = T_i, P = C1 Sa, K = \tau Sa$$

přepsat do finálního tvaru

```
> T = subs(su, rhs(T3));
```

$$T = T_e + C1 + (T_i - C1 - T_e) e^{-\frac{t}{\tau}}. \quad (5)$$

Určení zbytkového výkonu paliva

V případě, že naměříme teplotu stěny kontejneru v závislosti na čase a známe je možné pomocí regresní analýzy z rovnice (5) určit neznámé koeficienty $C1$ a τ . K výpočtu použijeme metodu nejmenších čtverců.

Předpokládáme, že v čase t_i jsme naměřili teplotu povrchu kontejneru Y_i . Celkový počet naměřených hodnot je N . Nejprve stanovíme druhou mocninu rozdílu dT_i mezi funkční hodnotou teploty $T(t_i)$ a odpovídající naměřenou teplotou Y_i . Pro zvýšení přehlednosti výpočtu použijeme substituce, které zjednoduší nejčastěji používané výrazy.

```
> dT[i] := subs(su1,su2,t=t[i],T)-Y[i]:
> dT2[i] := expand(dT[i]^2);
```

$$\begin{aligned} dt2_i := & C1^2 \left(e^{\frac{t_i}{\tau}}\right)^{-2} - 2 C1^2 \left(e^{\frac{t_i}{\tau}}\right)^{-1} - 4 C1 T e \left(e^{\frac{t_i}{\tau}}\right)^{-1} + 2 C1 Y_i \left(e^{\frac{t_i}{\tau}}\right)^{-1} \\ & + T i^2 \left(e^{\frac{t_i}{\tau}}\right)^{-2} + T e^2 \left(e^{\frac{t_i}{\tau}}\right)^{-2} - 2 T e^2 \left(e^{\frac{t_i}{\tau}}\right)^{-1} + C1^2 + 2 C1 T e - 2 C1 Y_i \\ & + T e^2 - 2 T e Y_i + Y_i^2 + 2 T e Y_i \left(e^{\frac{t_i}{\tau}}\right)^{-1} - 2 T i C1 \left(e^{\frac{t_i}{\tau}}\right)^{-2} - 2 T i T e \left(e^{\frac{t_i}{\tau}}\right)^{-2} \\ & + 2 T i C1 \left(e^{\frac{t_i}{\tau}}\right)^{-1} + 2 T i T e \left(e^{\frac{t_i}{\tau}}\right)^{-1} - 2 T i Y_i \left(e^{\frac{t_i}{\tau}}\right)^{-1} + 2 C1 T e \left(e^{\frac{t_i}{\tau}}\right)^{-2}. \end{aligned}$$

Proměnná $dt2_i$ je dosti nepřehledná, proto ve výpočtu sumy kvadrátu residuí, SQR, provedeme několik substitucí, které zvýší přehlednost výsledku.

```
> SQR := map(u->simplify(sum(u,i=1..N)),dT2[i]):
> sigma := selectfun(SQR,sum):
> Su := [seq(sigma[j]=Sigma[j],j=1..nops(sigma))];
> Bs := map(u->rhs(u)=lhs(u),Su):
> SQR := subs(Su,SQR);
```

$$Su := \left[\sum_{i=1}^N e^{-\frac{t_i}{\tau}} Y_i = \Sigma_1, \sum_{i=1}^N Y_i = \Sigma_2, \sum_{i=1}^N e^{-\frac{t_i}{\tau}} = \Sigma_3, \sum_{i=1}^N Y_i^2 = \Sigma_4, \sum_{i=1}^N e^{-2\frac{t_i}{\tau}} = \Sigma_5 \right]$$

$$\begin{aligned} SQR := & C1^2 \Sigma_5 - 2 C1^2 \Sigma_3 - 4 C1 T e \Sigma_3 + 2 C1 \Sigma_1 + T i^2 \Sigma_5 + T e^2 \Sigma_5 - 2 T e^2 \Sigma_3 \\ & + N C1^2 + 2 N C1 T e - 2 C1 \Sigma_2 + N T e^2 - 2 T e \Sigma_2 + \Sigma_4 + 2 T e \Sigma_1 \\ & - 2 T i C1 \Sigma_5 - 2 T e T i \Sigma_5 + 2 T i C1 \Sigma_3 + 2 T e T i \Sigma_3 - 2 T i \Sigma_1 + 2 C1 T e \Sigma_5. \end{aligned}$$

Nyní je nutné nalézt takovou hodnotu koeficientu $C1$, která minimalizuje sumu kvadrátu residuí SQR. Z této hodnoty je pak možné určit odpovídající výkon, případně provést zpětnou substituci, jejíž výstup zde není uveden z důvodů úspory místa.

```
> dC := diff(SQR,C1)=0:
> Vykon := (solve(su[1],C1)=collect(solve(dC,C1),[Te]))*S*a;
> Vykon := subs(Bs,tau=solve(su[2],tau),Vykon):
```

$$Vykon := P = S a \left(-T e - \frac{T i \Sigma_3 - \Sigma_2 - T i \Sigma_5 + \Sigma_1}{\Sigma_5 + N - 2 \Sigma_3} \right)$$

Jak je vidět, za předpokladu znalosti součinu $S a$, tepelné kapacity kontejneru K , počáteční T_i , konečné T_{inf} a teploty okolí T_e je možné pomocí proměření průběhu teploty povrchu kontejneru v čase určit zbytkový výkon paliva jako analytickou funkci naměřené teplotní závislosti.

Simulovaný ohřev

Předpokládejme, že sledujeme závislost teploty kontejneru naplněného vodou, který zahřívá zbytkový výkon uloženého paliva. Základní vlastnosti kontejneru popisuje tabulka 1.

V případě výše uvedených hodnot je možné graficky znázornit průběh teploty stěny kontejneru pro zbytkové výkony paliva od 5 kW do 30 kW s krokem 5 kW , viz obrázek 1.


Veličina	Proměnná	Velikost	Jednotka
Hmotnost obalu	Mo	75 000	kg
Hmotnost náplně	Mo	5 000	kg
Měrné teplo obalu	Co	520	$J\ kg^{-1}\ K^{-1}$
Měrné teplo vody	Cv	4186	$J\ kg^{-1}\ K^{-1}$
Povrch kontejneru	S	30	m^2
Koeficient přestupu tepla	alpha – α	25	$W\ m^{-2}\ K^{-1}$
Počáteční teplota	Ti	18	$^{\circ}C$
Teplota okolí	Te	25	$^{\circ}C$

Tab. 1. Základní proměnné

```

> K := Mv*Cv+Mo*Co: SA := S*alpha: tau := K/SA: C1 := P/SA:
> Mo := 75000: Co := 520: Mv := 5000: Cv := 4186: S := 30:
> alpha := 25: Ti := 18: Te := 25:
> PS := [5000,10000,15000,20000,25000,30000]:
> TF := map(u->evalf(subs(P=u,limit(T,t=infinity))),PS):
> P1 := plot({seq(T,P=PS)}, t=0..5*86400,
>           title="Predpokladany prubeh teploty",
>           labels=["t[s]", "T [C]"], color=black,thickness=3):
> P2 := plot(TF,t=0..5*86400,color=black,linestyle=4):
> P3 := plots[textplot]({seq([5*86400,TF[j]+1,
>           cat(convert(PS[j]/1000,string)," kW"]),j=1..6)},
>           align={ABOVE,LEFT}):
> plots[display]({P1,P2,P3});

```



Obr. 1. Průběh teploty stěny kontejneru pro různé výkony

Zpracování simulovaného měření

Předpokládejme zbytkový výkon paliva v kontejneru 20 kW. Parametry kontejneru uvažujeme stejné jako v tabulce 1. Dále předpokládejme, že měření teploty provádíme každou půlhodinu, časy měření jsou uloženy v seznamu `tt` a teploty povrchu kontejneru měříme s přesností na 0.5 °C, teploty uložíme v seznamu `Y`.

```
> P := 20000; chyba := rand(-10..10)*0.05:
> tt := [seq(1800*j, j=0..240)]:
> Y := map(u->evalf(subs(t=u+1800*40, T)+chyba(), 3), tt):
> P1 := plot(zip((u,v)->[u,v], tt, Y), symbol=circle,
>             style=point, title="Simulovane teploty",
>             color=black, symbolsize=15,
>             labels=["cas [s]", "t [°C]"]):
```

Proměnná `P1` obsahuje časový průběh simulovaných hodnot. Graf je součástí obrázku 2.

Data proložíme obecnou exponenciální funkcí $F = k_1 e^{-\frac{t}{\tau}} + K_2$. Funkce F obsahuje neznámý nelineární parametr τ , neznámý lineární parametr k_1 je v součinu s nelineární funkcí τ . Tato skutečnost vylučuje možnost určit parametry k_1 , τ , K_2 funkce F pomocí lineární regrese – metody nejmenších čtverců. Použijeme proto modifikovanou nelineární regresní iterační metodu odvozenou Gaussem a Newtonem, upravenou Ganderem a Bartoňem [2, 3]. Z těchto parametrů později určíme základní charakteristiku problému – zbytkový výkon paliva P .

Prvním krokem je linearizace funkce podle proměnných k_1 a τ v okolí počátečních hodnot K_1 a T .

```
> unassign('Tau', 'tau', 'K1', 'T0');
> F := exp(-t/tau)*k1+K2:
> e1 := mtaylor(F, [tau=Tau, k1=K1], 2);
```

$$e1 := e^{-\frac{t}{T}} K_1 + K_2 + e^{-\frac{t}{T}} (k_1 - K_1) + \frac{e^{-\frac{t}{T}} (\tau - T) t K_1}{T^2}.$$

Nyní nahradíme rozdíly $k_1 - K_1 = DK$, $\tau - T = DT$ opravami k počátečním hodnotám.

```
> Dsu := [k1-K1=DK, tau-Tau=DT]:
> e2 := subs(Dsu, e1);
```

$$e2 := e^{-\frac{t}{T}} K_1 + K_2 + e^{-\frac{t}{T}} DK + \frac{e^{-\frac{t}{T}} DT t K_1}{T^2}.$$

Proměnná `e2` je již lineární vzhledem k opravám DK a DT pro počáteční hodnoty K_1 a T . Dále je samozřejmě lineární pro K_2 . Vytvoříme seznam koeficientů u jednotlivých lineárních proměnných a dále vybereme ten člen, který lineární koeficienty neobsahuje.

```
> Var := [DK, DT, K2];
> Col := map(u->select(has, e2, u)/u, Var);
> rest := remove(has, e2, Var);
```

$$\text{Col} := \left[e^{-\frac{t}{T}}, \frac{e^{-\frac{t}{T}} t K1}{T^2}, 1 \right]$$

$$\text{rest} := e^{-\frac{t}{T}} K1$$

Nyní zbývá zvolit výchozí hodnoty $K1$ a T . Pokud je dosadíme do proměnné Col a za čas t budeme postupně dosazovat číselné hodnoty ze seznamu tt dostaneme numerickou matici $\mathbf{A}$. Pokud provedeme totéž s proměnnou rest získáme seznam rozdílů naměřených hodnot $\mathbf{Y}$ a funkčních hodnot funkce F pro jednotlivé časy ze seznamu tt . Při správné volbě výchozích hodnot $K1$ a T budou tyto rozdíly malé. Opravy DK a DT pro výchozí hodnoty $K1$ a T a koeficient $K2$ minimalizují následující výraz, viz příkaz MAPLE `leastsqrs` [4].

$$\min_{[DK, DT, K2]} \|\mathbf{A} \bullet [DK, DT, K2] - [\text{rest}]\|_2$$

Vypočtené opravy DK a DT pak přičteme k výchozím hodnotám $K1$ a T a celý postup znovu zopakujeme. Výpočet – iteraci ukončíme v okamžiku, kdy absolutní velikost opravy klesne pod stanovenou mez. V průběhu iterace zvýšíme numerickou přesnost na 24 platných cifer.

```
> K1 := 30; Tau := 75000;
> Digits := 24; epsilon := 10; n := 0;
> while epsilon > 10^(-Digits/2) do;
>   n := n+1;
>   A := matrix(evalf(map(u->subs(t=u, Col), tt)));
>   B := evalf(zip((u,v)->u-subs(t=v, rest), Y, tt));
>   SOL := linalg[leastsqrs](A,B);
>   epsilon:=max(map(u->abs(u), convert(SOL, list)[1..2]))[];
>   if epsilon<1 then LM:=1 end if;
>   K1:=K1+SOL[1]/LM; Tau:=Tau+SOL[2]/LM;
>   print(krok=n, 'epsilon'=epsilon);
> end do;
```


```
krok = 1, ε = 1586.57932207589427114091
krok = 2, ε = 1953.34519717514051488873
krok = 3, ε = 50.2249896781355783584539
krok = 4, ε = 0.747346442956547593010003
krok = 5, ε = 0.0109678001010463140302263
krok = 6, ε = 0.000160891760886171317598603
krok = 7, ε = 0.236018114804541723999455 10-5
krok = 8, ε = 0.346223726241167823815361 10-7
krok = 9, ε = 0.507888411927475432324407 10-9
krok = 10, ε = 0.745036077464870534422161 10-11
krok = 11, ε = 0.109364870794910660862430 10-12.
```

Po ukončení iterace se vrátíme ke standardní numerické přesnosti, vyjádříme výslednou funkci popisující průběh teploty v čase a vykreslíme ji spolu s naměřenými hodnotami, viz obrázek 2.

```
> Digits := 10;
> F := evalf(subs(tau=Tau, k1=K1, K2=SOL[3], F));
> Finf := limit(F, t=infinity);
> P2 := plot([F, Finf], t=0..max(tt[]*1.1), thickness=[5, 1],
>          linestyle=[1, 3], color=red);
> plots[display]({P1, P2});
```

$$F := -13.72950375 e^{-0.00001272411817 t} + 51.64362522$$

$$F_{\infty} := 51.64362522$$



Obr. 2. Průběh teploty jako regresní funkce naměřených hodnot

Pokud nyní porovnáme vypočtené koeficienty s obecnými koeficienty z rovnice (5) je možné snadno dopočítat počáteční teplotu t_i , výkon p a koeficient sa – zde jsou na rozdíl od rovnice (5) označeny malými písmeny. Po dosazení $k = K$ a $te = Te$ dostaneme konkrétní číselné hodnoty.

```
> e1 := Tau=k/sa: e2 := K1=ti-p/sa-te: e3 := SOL[3]=p/sa+te:
> Sol := solve({e3, e2, e1}, {ti, p, sa});
> Sol := subs(k=K, te=Te, Sol);
```

$$\text{Sol} := \{ ti = 37.91412146, sa = 0.00001272411817 k, \\ p = 0.0006571195898 k - 0.00001272411817 te k \}$$

$$\text{Sol} := \{ sa = 762.5564019, p = 20317.26697, ti = 37.91412146 \} .$$

Porovnáme – li vypočtený výkon s předpokládanou hodnotou, zjistíme, že chyba výsledku je přibližně 1.6%, což je velmi slušná shoda.

Nyní stanovíme korelační koeficient *Korelace* mezi naměřenými hodnotami a funkčními hodnotami regresní funkce *F*, dále stanovíme průměrnou odchylku *pd_y* mezi naměřenou teplotou a funkční hodnotou funkce *F* a stanovíme střední kvadratickou chybu *sdy* této odchylky.

```
> assign(Sol
> y := map(u->evalf(subs(t=u,F)),tt):
> Korelace:=stats[describe,linearcorrelation](y,Y);
> dy := zip((u,v)->u-v,y,Y):
> N := nops(tt):
> pd y:= sum(dy[i],i=1..N)/N;
> sdy := sqrt(sum((dy[i]-pd y)^2,i=1..N)/N);
```

$$\begin{aligned}Korelace &:= .9956710206 \\pd y &:= .3858921162 \cdot 10^{-8} \\sdy &:= .3122672182\end{aligned}$$


Vidíme, že průměrná odchylka *pd_y* je zcela zanedbatelná a střední kvadratická chyba *sdy* je přibližně dvě třetiny chyby, kterou byly zatíženy vstupní hodnoty, viz Obrázek 3. Korelační koeficient blízký jedné vypovídá o vynikající shodě regresní funkce s daty, viz Obrázek 4.

```
> plot([zip((u,v)->[u/1000,v],tt,dy),
> [[0,pd y],[tt[N]/1000,pd y]], [[0,pd y+sdy],
> [tt[N]/1000,pd y+sdy]], [[0,pd y-sdy],
> [tt[N]/1000,pd y-sdy]]],style=[point,line,line,line],
> thickness=[1,3,3,3],linestyle=[1,1,3,3],
> labels=["t [s]", "DT [C]"],symbol=circle,
> color=[black,red,red,red],symbolsize=15,
> title="Znazorneni chyby vypoctu",
> labels=["t [s]", "T [C]"]);
> plot(zip((u,v)->[u,v],Y,y),color=black,
> labels=["Y [C]", "y [C]"],
> title="Znazorneni kvality korelace");
```


Závěr

Z výše uvedených výpočtů vyplývá, že měřením teploty povrchu kontejneru s vyhořelým jaderným palivem je možné určit zbytkový výkon paliva. Metodu je však možné použít pokud známe tepelnou kapacitu naplněného kontejneru, kterou je možné stanovit buď z technické dokumentace, nebo měřením.

Stanovení zbytkového výkonu z naměřeného průběhu teploty vyžaduje konstantní teplotu a způsob proudění chladicího media, (voda, vzduch) v okolí kontejneru. Tuto podmínky lze ale v prostředí úložiště vyhořelého paliva snadno zajistit.



Obr. 3. Znázornění chyby výpočtu



Obr. 4. Znázornění kvality korelace

Použitá literatura

- [1] Incropera F. P., DeWitt D. P.: *Fundamentals of Heat and Mass transfer*. John Wiley, New York, 1996, ISBN 0-471-3046+-0-3
- [2] Überhuber Ch.: *Computer–Numerik*. Springer Verlag, Berlin, 1995, ISBN 3-540-59152-4
- [3] Gander W., Bartoň S.: *Least Squares Fit with Piecewise Functions*. In Gander W., Hřebíček J.: *Solving Problems in Scientific Computing Using MAPLE and MATLAB*. Springer Verlag, Heidelberg, 2004, ISBN 3-540-21127-6
- [4] *Maple 10, User Manual*. Maplesoft, Waterloo, 2005, ISBN 1-894511-75-1